

Mplus analysis code for:

Big-Fish-Little-Pond Effects on Ninth-Grade Students' Mathematics and Language Self-Concepts: The Moderating Role of Cognitive Ability

TITLE: Model 1, basic BFLPE model

DATA: File is bflpe.dat;

VARIABLE: Names are RS9SCH NormWgt6 MaGrade MSC1 MSC2 MSC3;

Usevariables are MaGrade MSC1 MSC2 MSC3;

Cluster = RS9SCH;

Weight is NormWgt6;

WTSCALE=UNSCALED;

Missing are all (-999);

DEFINE: STANDARDIZE MaGrade MSC1 MSC2 MSC3;

ANALYSIS:

Type = twolevel;

Estimator = MLR;

MODEL:

%WITHIN%

MSC_w BY MSC1 (1);

MSC_w BY MSC2 (2);

MSC_w BY MSC3 (3);

MSC_w ON MaGrade (b_within);

MaGrade (Psi_W);

MSC_w (Theta_W);

%BETWEEN%

MSC_b BY MSC1 (1);

MSC_b BY MSC2 (2);

MSC_b BY MSC3 (3);

MSC_b ON MaGrade (b_betwn);

MaGrade (Psi_B);

MSC_b (Theta_B);

MODEL CONSTRAINT:

new(bflpe);

bflpe = b_betwn - b_within;

new(ES2);

Es2= bflpe*(2*sqrt(Psi_B)/sqrt(Psi_W*b_within**2 + Theta_W));

OUTPUT: Stdyx;

TITLE: Model 2, BFLPE model with direct effects of gender and cogn

DATA: File is bflpe.dat;

VARIABLE: Names are RS9SCH NormWgt6 MaGrade gender cogn MSC1 MSC2 MSC3;

Usevariables are MaGrade gender cogn MSC1 MSC2 MSC3;

Cluster = RS9SCH;

Weight is NormWgt6;

WTSCALE=UNSCALED;

Missing are all (-999);

Within = gender cogn;

DEFINE: STANDARDIZE MaGrade gender cogn MSC1 MSC2 MSC3;

ANALYSIS:

Type = twolevel;

Estimator = MLR;

MODEL:

%WITHIN%

MSC_w BY MSC1 (1);

MSC_w BY MSC2 (2);

MSC_w BY MSC3 (3);

MSC_w ON MaGrade (b_within);

MSC_w ON cogn;

MSC_w ON gender;

MaGrade with cogn;

MaGrade (Psi_W);

MSC_w (Theta_W);

%BETWEEN%

MSC_b BY MSC1 (1);

MSC_b BY MSC2 (2);

MSC_b BY MSC3 (3);

MSC_b ON MaGrade (b_betwn);

MaGrade (Psi_B);

MSC_b (Theta_B);

MODEL CONSTRAINT:

new(bflpe);

bflpe = b_betwn - b_within;

new(ES2);

Es2= bflpe*(2*sqrt(Psi_B)/sqrt(Psi_W*b_within**2 + Theta_W));

OUTPUT: Stdyx;

TITLE: Model 3, BFLPE model with cross-level interaction

DATA: File is bflpe.dat;

VARIABLE: Names are RS9SCH NormWgt6 MaGrade gender cogn MSC1 MSC2 MSC3;

Usevariables are MaGrade gender cogn MSC1 MSC2 MSC3;

Cluster = RS9SCH;

Missing are all (-999);

Within = gender cogn;

DEFINE: STANDARDIZE MaGrade gender cogn MSC1 MSC2 MSC3;

ANALYSIS:

Type = twolevel random;

Estimator = BAYES;

H1ITERATIONS = 10000;

MODEL:

%WITHIN%

MSC_w BY MSC1 (1);

MSC_w BY MSC2 (2);

MSC_w BY MSC3 (3);

MSC_w ON gender;

s | MSC_w ON cogn;

MSC_w ON MaGrade (b_within);

MaGrade with cogn;

MaGrade (Psi_W);

MSC_w (Theta_W);

%BETWEEN%

MSC_b BY MSC1 (1);

MSC_b BY MSC2 (2);

MSC_b BY MSC3 (3);

MSC_b ON MaGrade (b_betwn);

s ON MaGrade;

s with MSC_b;

MaGrade (Psi_B);

MSC_b (Theta_B);

MODEL CONSTRAINT:

new(bflpe);

bflpe = b_betwn - b_within;

new(ES2);

Es2= bflpe*(2*sqrt(Psi_B)/sqrt(Psi_W*b_within**2 + Theta_W));

OUTPUT: Stdyx;